
This section discusses the structure of standard maps and how to create and use a
user-defined map.  A standard map is a map supplied with NonStop NET/MASTER
MS to access data structures that are an integral part of NonStop NET/MASTER MS.
A user-defined map is a map that you create to access data structures that you define.

Table 11-1 summarizes the verbs discussed in this section.  These are verbs that
support access to Mapping Services and data in MDO variables.

Table 11-1.  Verbs Used With Maps and MDO Variables

Verbs Description

ASSIGN verb Manipulates multiple variables in a single operation.

DROP verb Deletes (drops) multiple variables in a single operation.

EMS* verbs Controls the flow of Event Management Service (EMS) messages through the
system-level EMSPROC NCL procedure.

FILE verbs Opens and closes user databases (UDBs) and manipulates UDB records.

INT* verbs Controls the flow of messages sent from dependent processing environments.

LOG* verbs Controls the flow of messages through the system-level LOGPROC NCL
procedure.  These are messages destined for the NonStop NET/MASTER MS
activity log.

MSG* verbs Controls the flow of messages through the system-level MSGPROC NCL
procedure.  These are messages destined for the OCS window with which the
MSGPROC procedure is associated.

SECCALL verbs Manipulates user ID definition records in the User ID Management Services
(UMS) database.

WRITE verb Writes a message to a terminal, an NCL process, all monitor-class users, a
specified user, or the activity log.

INT*, EMS*, LOG*, and MSG* refer to all verbs beginning with the letters INT, EMS, LOG, and MSG,
respectively.

For the complete syntax of verbs, refer to the NonStop NET/MASTER NCL Reference
Manual.  (In this manual, Section 2, “NCL Summary,” discusses the meaning of the
asterisk (*) following the letters EMS, INT, LOG, and MSG.)

Table 11-2 summarizes the standard maps discussed in this section.

Table 11-2.  Standard Maps

Map Description

$MSG Used to interpret the data in an MDO variable that contains a message traveling
to, from, or within NonStop NET/MASTER MS.

$NCL Used to interpret the data in an MDO variable that contains a record being written
to or read from a mapped user database (UDB).

$SEC Used to interpret the data in an MDO variable that contains a record being written
to or read from the User ID Management Services (UMS) database.

$SPI Used to interpret the portion of the data in an MDO variable mapped by the map
$MSG that contains the NonStop NET/MASTER MS equivalent of Subsystem
Programmatic Interface (SPI) tokens.

Table 11-3 summarizes the verbs that can refer to standard maps.

Table 11-3.  Maps Referred to by Verbs

Map Verbs

$MSG ASSIGN, EMS*, INT*, LOG*, MSG*

$NCL ASSIGN, FILE, WRITE

$SEC ASSIGN, SECCALL

$SPI ASSIGN, EMS*, INT*, LOG*, MSG* (Verbs do not usually refer directly to $SPI.)

INT*, EMS*, LOG*, and MSG*, refer to all verbs beginning with the letters INT, EMS, LOG, and MSG,
respectively.

(Section 2, “NCL Summary,” discusses the meaning of the asterisk (*) following the
letters EMS, INT, LOG, and MSG.)

The Standard Map
$NCL

The standard map $NCL is used by the FILE and WRITE verbs.  Figure 11-1 shows the
structure of the map $NCL.  It shows that data structured by this map corresponds to
the following MDO variable:

&stem-name.DATALIST

stem-name

specifies the name of the stem of the MDO variable that holds the record.

Fields in data structured by this map correspond to the following repeated MDO
variables:

…
&stem-name.DATALIST.DATA{1}
&stem-name.DATALIST.DATA{2}
&stem-name.DATALIST.DATA{3}
…

The FILE Verbs and the
Map $NCL

The FILE verbs—FILE ADD, FILE CLOSE, and so on—open and close user databases
(UDBs) and manipulate UDB records.  The map $NCL is used to interpret the data in
an MDO variable that contains a record being written to or read from a mapped UDB.
(The FILE GET, FILE OPEN, and FILE PUTGET verbs allow you to use the MAP
operand to specify a different map if you wish, but this is not recommended.)

The FILE verbs are discussed in more detail in Section 12, “Working With Files.”

The WRITE Verb and the
Map $NCL

The WRITE verb uses the map $NCL when sending data between NCL processes in
ordinary variables using either the ARGS or VARS operand.  The data is sent as an
enclosed MDO variable, which is mapped by the map $NCL, in a message, which is
mapped by the map $MSG.

This use of the WRITE verb is discussed in Section 16, “Environments and Command
Processing.”

Figure 11-1.  The Map $NCL

The Standard Map
$MSG

The standard map $MSG is used by the message verbs.  The message verbs—EMS*,
INT*, LOG*, and MSG*—control the flow of messages traveling to, from, and within
NonStop NET/MASTER MS.  The map used by these verbs to interpret the data in an
MDO variable that contains a message is $MSG.  (Section 2, “NCL Summary,”
discusses the meaning of the asterisk (*) following the letters EMS, INT, LOG, and
MSG.)

Note Some MDO variables mapped by the map $MSG may contain the NonStop NET/MASTER MS equivalent
of Subsystem Programmatic Interface (SPI) tokens.  The map $SPI is used to interpret the portion of the
data in an MDO variable that contains the NonStop NET/MASTER MS equivalent of SPI tokens.  The
map $SPI is discussed later in this section.

Figure 11-2 shows the structure of the map $MSG.  An actual MDO variable containing
a message does not have all the elements shown in this figure.  For example:  if the
entire text of a message is displayed with the same attributes, the MDO variable does
not contain the FLD element;  if the message does not contain SPI tokens, the MDO
variable does not contain the SPI element;  if the message is not a PPO message, the
MDO variable does not contain the PPO element;  and so on.

The name of the MDO variable that contains the message data depends on the name of
the verb that reads the message.  This is summarized in the following table:

Verb MDO Variable Example Containing Base Color Attribute

EMSREAD &$EMS. &$EMS.MSGATTR.DISPLAY.COLOR

INTREAD &$INT. &$INT.MSGATTR.DISPLAY.COLOR

LOGREAD &$LOG. &$LOG.MSGATTR.DISPLAY.COLOR

MSGREAD &$MSG. &$MSG.MSGATTR.DISPLAY.COLOR

The MDO variables listed in the preceding table are automatically created by the
EMSREAD, INTREAD, LOGREAD, and MSGREAD verbs.  You do not have to create
these variables using the ASSIGN verb before reading a message.

Data is automatically assigned to these variables when a verb reads a message.  You
can refer to element, group, and field names to obtain the contents of a message as the
following examples show:

&$EMS.MSGATTR.DISPLAY.COLOR           /* Base COLOR field  */
&$EMS.MSGATTR.DISPLAY.HLITE           /* Base HLITE field  */
&$EMS.MSGATTR.DISPLAY.INTENS          /* Base INTENS field */
&$INT.MSGATTR.FLD{1}.OFFSET           /* 1st OFFSET field  */
&$INT.MSGATTR.FLD{1}.LENGTH           /* 1st LENGTH field  */
&$INT.MSGATTR.FLD{1}.COLOR            /* 1st COLOR field   */
&$INT.MSGATTR.FLD{1}.HLITE            /* 1st HLITE field   */
&$INT.MSGATTR.FLD{1}.INTENS           /* 1st INTENS field  */

The message verbs are discussed in more detail in Section 16, “Environments and
Command Processing,” and Section 17, “Developing System-Level NCL Procedures.”

The values that a message profile variable can contain are listed in the NonStop
NET/MASTER NCL Reference Manual.

Figure 11-2.  The Map $MSG (Page 1 of 5)

Figure 11-2.  The Map $MSG (Page 2 of 5)

Figure 11-2.  The Map $MSG (Page 3 of 5)

Figure 11-2.  The Map $MSG (Page 4 of 5)

Figure 11-2.  The Map $MSG (Page 5 of 5)

The Standard Map
$SPI

The standard map $MSG is used by the message verbs.  Event Management Service
(EMS) event messages arriving at NonStop NET/MASTER MS, from Tandem
subsystems such as Expand, Pathway, and SNAX/CDF, always contain SPI tokens.

When an event message arrives at NonStop NET/MASTER MS, NonStop
NET/MASTER MS creates two structures.  It creates one structure containing the text
of the message that Mapping Services can interpret using the map $MSG, and it
creates another structure containing the NonStop NET/MASTER MS equivalent of SPI
tokens that Mapping Services can understand using the map $SPI.

Both structures are present in the same MDO variable after it is created, but NonStop
NET/MASTER MS uses the map $MSG to interpret the text part of the MDO variable
and the map $SPI to interpret the SPI part of the MDO variable.

This subsection does not present the map $SPI in a figure because it would be too
large—many Tandem subsystems use SPI tokens.  Rather, this subsection presents the
general rules you can use to detect and interpret messages that contain SPI tokens and
the rules you can use to construct messages that contain SPI tokens.

Refer to the Introduction to Distributed Systems Management (DSM) and the Distributed
Systems Management (DSM) Programming Manual manuals for initial information on
SPI tokens.  Refer to the manuals that describe individual subsystems for detailed
information on the SPI tokens used by each subsystem.

Note Some NonStop NET/MASTER MS messages are EMS events that contain SPI tokens.  These messages
are listed in the NonStop NET/MASTER Messages Manual.

SPI to MDO Conversion
Rules

When an event message containing SPI tokens arrives at NonStop NET/MASTER MS,
NonStop NET/MASTER MS converts the SPI tokens to a structure that Mapping
Services can understand using the map $SPI.  It converts the SPI tokens using a series
of rules.

SPI to MDO Conversion Rule 1

NonStop NET/MASTER MS makes an EMSTEXT system call.  The text of the message
preceded by the message number is placed in the text part of the MDO variable,
&$EMS.TEXT.

SPI to MDO Conversion Rule 2

NonStop NET/MASTER MS assumes that all SPI token names—across all Tandem
subsystems that use SPI tokens—are unique, and converts each SPI token in turn
according to the following conversion rules:

SPI to MDO Conversion Rule 2a.  NonStop NET/MASTER MS converts all hyphens (-) in a
SPI token name to underscores (_) in an MDO variable name.  The following table lists
some SPI token names before conversion and their corresponding MDO variable
names after conversion in a message that is read by the MSGREAD verb:

SPI Token Name
MDO Variable Name
(&$MSG.SPI.TANDEM. begins all variable names.)

ZSPI-TKN-SSID ZSPI_TKN_SSID

ZEMS-TKN-CPU ZEMS_TKN_CPU

ZEMS-TKN-CRTPID ZEMS_TKN_CRTPID

ZEMS-TKN-EMPHASIS ZEMS_TKN_EMPHASIS

ZEMS-TKN-TEXT ZEMS_TKN_TEXT

ZEMS-TKN-SUBJECT-MARK ZEMS_TKN_SUBJECT_MARK

SPI to MDO Conversion Rule 2b.  NonStop NET/MASTER MS qualifies all converted SPI
tokens by the owner of the subsystem that defines the token and groups together with
all tokens owned by a specific owner in an MDO variable.  The following table lists
some SPI token names before conversion and their corresponding MDO variable
names after conversion in a message that is read by the MSGREAD verb:

SPI Token Name
MDO Variable Name
(&$MSG.SPI. begins all variable names.)

TANDEM.EMS.D30 ZEMS-TKN-TEXT TANDEM.ZEMS_TKN_TEXT

USER.0.0 USER-TKN-ABC USER.USER_TKN_ABC

SPI to MDO Conversion Rule 2c.  NonStop NET/MASTER MS converts all subordinate
fields in a SPI token to fields that are subordinate to the SPI token name in an MDO
variable.  (The SPI token name is effectively regarded as an element name.)  The
following table shows how a SPI token that has subordinate fields—ZEMS-MAP-
STATUS-DIST—appears after conversion into an MDO variable in a message that is
read by the MSGREAD verb:

SPI Token Name Subordinate Field
MDO Variable Name
(&$MSG.SPI.TANDEM. begins all variable names.)

ZEMS-MAP-STATUS-DIST ZDIST-NAME ZEMS_MAP_STATUS_DIST.ZDIST_NAME

ZDIST-TYPE ZEMS_MAP_STATUS_DIST.ZDIST_TYPE

ZDIST-PRIMARY-CPU ZEMS_MAP_STATUS_DIST.ZDIST_PRIMARY_CPU

ZDIST-BACKUP-CPU ZEMS_MAP_STATUS_DIST.ZDIST_BACKUP_CPU

SPI to MDO Conversion Rule 2d.  NonStop NET/MASTER MS converts all subordinate
tokens in a SPI token list to elements that are subordinate to the SPI token name in an
MDO variable.  (The SPI token name is effectively regarded as an element name.)  The
following table shows how a SPI token that has subordinate tokens in a list—ZSPI-
TKN-ERRLIST—appears after conversion into an MDO variable in a message that is
read by the MSGREAD verb:

SPI Token Name Token in List
MDO Variable Name
(&$MSG.SPI.TANDEM. begins all variable names.)

ZSPI-TKN-ERRLIST ZSPI-TKN-ERROR ZSPI_TKN_ERRLIST.ZSPI_TKN_ERROR

ZSPI-TKN-PARM-ERR ZSPI_TKN_ERRLIST.ZSPI_TKN_PARM_ERR

ZSPI-TKN-ERRLIST ZSPI_TKN_ERRLIST.ZSPI_TKN_ERRLIST

Note The ZSPI-TKN-ENDLIST token is filtered by NonStop NET/MASTER MS and is not present in an MDO
variable.

SPI to MDO Conversion Rule 2e.  NonStop NET/MASTER MS adds a period (.) and
number sign (#) field to each token of type enumeration to get the unenumerated
value of the token.  NonStop NET/MASTER MS adds a number sign field to each field
of type enumeration to get the unenumerated value of the field.  The plain name of the
token or field—without the number sign—gets the enumerated value—the symbolic
name.  Hyphens are converted to underscores in an enumerated value.

The following table shows enumerated and unenumerated values for some event
numbers and symbolic names defined by the Expand subsystem in a message that is
read by the MSGREAD verb:

Expand Event Number Value
MDO Variable Name
(&$MSG.SPI.TANDEM. begins all variable names.)

EXP0003 3
ZEXP_ENM_IO_BUS_ERR

ZEMS_TKN_EVENTNUMBER.# (unenumerated value)
ZEMS_TKN_EVENTNUMBER (enumerated value)

EXP0004 4
ZEXP_ENM_DEV_ERR

ZEMS_TKN_EVENTNUMBER.# (unenumerated value)
ZEMS_TKN_EVENTNUMBER (enumerated value)

EXP0006 6
ZEXP_ENM_DEV_UP

ZEMS_TKN_EVENTNUMBER.# (unenumerated value)
ZEMS_TKN_EVENTNUMBER (enumerated value)

EXP0007 7
ZEXP_ENM_DEV_DOWN

ZEMS_TKN_EVENTNUMBER.# (unenumerated value)
ZEMS_TKN_EVENTNUMBER (enumerated value)

Note Not all Tandem subsystems have enumerated event numbers.  For this reason, unless you are certain
that a subsystem has enumerated event numbers, it is best to use the unenumerated form of an MDO
variable name.  For example:

&$MSG.SPI.TANDEM.ZEMS_TKN_EVENTNUMBER.# = 9999
&abc = &$MSG.SPI.TANDEM.ZEMS_TKN_EVENTNUMBER.#

Refer to the Data Definition Language (DDL) Reference Manual for more information on defining tokens
and DDL data types.

SPI to MDO Conversion Rule 2f.  Every EMS message has at least one subject.  The token
ZEMS-TKN-SUBJECT-MARK marks each subject in a message.  In an MDO variable,
the subject mark becomes the name of the subject token, including the occurrence
number since the message can have more than one subject.

You can obtain the name of the subject of a message using the following MDO
variable:

&$xxx.SPI.TANDEM.ZEMS_TKN_SUBJECT_MARK

The following example assigns the name of the subject of a message read by the
MSGREAD verb into the simple variable &A:

&a = &$MSG.SPI.TANDEM.ZEMS_TKN_SUBJECT_MARK

For example, the name of the subject of a message could be the following:

TANDEM.ZCOM_TKN_SUBJ_LINE{1}

You can obtain the value of the subject of a message two ways:  in a single assignment
statement or using two assignment statements.

The first method, using a single assignment statement, uses a nested MDO variable.
The following example assigns the value of the subject of a message read by the
MSGREAD verb into the simple variable &B:

&b = &$MSG.SPI.&($MSG.SPI.TANDEM.ZEMS_TKN_SUBJECT_MARK)



The Standard Map $SPI

Standard and User-Defined Maps

106160 Tandem Computers Incorporated 11–15

The second method, using two assignment statements, uses two simple variables.  The
first assignment statement assigns the name of the subject of a message read by the
MSGREAD verb into the simple variable &A.  The second assignment statement uses
&A in an MDO variable to then assign the value of the subject into the simple variable
&B:

&a = &$MSG.SPI.TANDEM.ZEMS_TKN_SUBJECT_MARK
&b = &$MSG.SPI.&a

For example, the value of the subject of a message could be the following:

\SYS1.$SNA05

SPI to MDO Conversion Rule 2g.  NonStop NET/MASTER MS converts the values of some
SPI tokens and fields from internal form to external form before being stored in an
MDO variable.  Table 11-4 lists the SPI tokens and fields that are converted to external
form.  It also lists the Guardian system procedure call used to convert a token to
external form from internal form (and to internal form from external form).

Refer to the Guardian Procedure Calls Reference Manual, Volume 1 and the Guardian
Procedure Calls Reference Manual, Volume 2 for a detailed description of all Guardian
system procedure calls.

Table 11-4.  SPI Tokens Converted to External Form in an MDO Variable (Page 1 of 3)

SPI Tokens and Fields Conversion Method Used Result of Conversion to External Form

ZSPI-TDT-FNAME Converted to external form using the Guardian
FNAMECOLLAPSE procedure call.
(FNAMEEXPAND converts to internal form.)

$volume.subvol.filename (local filename) or
\sys.$volume.subvol.filename (network filename)

ZSPI-TDT-FNAME32 Converted to external form using the Guardian
FNAME32COLLAPSE procedure call.
(FNAME32EXPAND converts to internal form.)

\sys.$volume.subvol.filename

ZSPI-TDT-DEVICE Converted to external form using the Guardian
FNAMECOLLAPSE procedure call.
(FNAMEEXPAND converts to internal form.)

\sys.$device

ZSPI-TDT-CRTPID Converted to external form using the Guardian
FNAMECOLLAPSE procedure call.
(FNAMEEXPAND converts to internal form.)

\sys.$process-name cpu,pin or
\sys.cpu,pin cpu,pin

ZSPI-TDT-SUBVOL Converted to external form using the Guardian
FNAMECOLLAPSE procedure call.
(FNAMEEXPAND converts to internal form.)

\sys.$volume.subvol

ZSPI-TDT-TIMESTAMP Converted to the following time format:
yyyy/mm/dd hh:mm:ss.ddd (in text).

yyyy/mm/dd hh:mm:ss.ddd

ZSPI-TDT-SSID Converted to external form using the Guardian
SSIDTOTEXT procedure call.  (TEXTTOSSID
converts to internal form.)

owner.ss.version or
0.0.0

ZSPI-TDT-TRANSID Converted to external form using the Guardian
TRANSIDTOTEXT procedure call.
(TEXTTOTRANSID converts to internal form.)

\sys-name(crash-count).cpu.sequence,
\sys-number(crash-count).cpu.sequence,
\sys-name.cpu.sequence,
\sys-number.cpu.sequence or
cpu.sequence

ZSPI-TDT-FLT Converted to NCL floating point format. number.number

ZSPI-TDT-INT4 Converted to NCL floating point format. number.number

ZSPI-TDT-FLT2 Converted to NCL floating point format. number.number

ZSPI-TDT-BOOLEAN Converted to 0 or 1. 0 or 1

ZSPI-TDT-INT Not converted to external form unless
enumerated and in a shared token, in which case
it is converted to long with the high 16 bits as the
subsystem number.  (Removing the high 16 bits
converts to internal form.)

N.A.

ZSPI-TDT-INT2 Not converted to external form unless
enumerated and in a shared token, in which case
it is converted to long with the high 16 bits as the
subsystem number.  (Removing the high 16 bits
converts to internal form.)

N.A.

N.A. Not applicable

Table 11-4.  SPI Tokens Converted to External Form in an MDO Variable (Page 2 of 3)

SPI Tokens and Fields Conversion Method Used Result of Conversion to External Form

ZSPI-TDT-UINT Not converted to external form unless
enumerated and in a shared token, in which case
it is converted to long with the high 16 bits as the
subsystem number.  (Removing the high 16 bits
converts to internal form.)

N.A.

ZSPI-TDT-ENUM Not converted to external form unless
enumerated and in a shared token, in which case
it is converted to long with the high 16 bits as the
subsystem number.  (Removing the high 16 bits
converts to internal form.)

N.A.

ZSPI-TDT-STRUCT Fields are converted to external form using the
data types described in the DATA-TYP- rows.

N.A.

DATA-TYP-FNAME Converted to external form using the Guardian
FNAMECOLLAPSE procedure call.
(FNAMEEXPAND converts to internal form.)

$volume.subvol.filename (local filename) or
\sys.$volume.subvol.filename (network filename)

DATA-TYP-FNAME32 Converted to external form using the Guardian
FNAME32COLLAPSE procedure call.
(FNAME32EXPAND converts to internal form.)

\sys.$volume.subvol.filename

DATA-TYP-DEVICE Converted to external form using the Guardian
FNAMECOLLAPSE procedure call.
(FNAMEEXPAND converts to internal form.)

\sys.$device

DATA-TYP-CRTPID Converted to external form using the Guardian
FNAMECOLLAPSE procedure call.
(FNAMEEXPAND converts to internal form.)

\sys.$process-name cpu,pin or
\sys.cpu,pin cpu,pin

DATA-TYP-SUBVOL Converted to external form using the Guardian
FNAMECOLLAPSE procedure call.
(FNAMEEXPAND converts to internal form.)

\sys.$volume.subvol

DATA-TYP-TSTAMP Converted to the following time format:
yyyy/mm/dd hh:mm:ss.ddd (in text).

yyyy/mm/dd hh:mm:ss.ddd

DATA-TYP-SSID Converted to external form using the Guardian
SSIDTOTEXT procedure call.  (TEXTTOSSID
converts to internal form.)

owner.ss.version or
0.0.0

DATA-TYP-TRANSID Converted to external form using the Guardian
TRANSIDTOTEXT procedure call.
(TEXTTOTRANSID converts to internal form.)

\sys-name(crash-count).cpu.sequence or
\sys-number(crash-count).cpu.sequence or
\sys-name.cpu.sequence or
\sys-number.cpu.sequence or
cpu.sequence

DATA-TYP-FLOAT32 Converted to NCL floating point format. number.number

N.A. Not applicable

Table 11-4.  SPI Tokens Converted to External Form in an MDO Variable (Page 3 of 3)

SPI Tokens and Fields Conversion Method Used Result of Conversion to External Form

DATA-TYP-INT4 Converted to NCL floating point format. number.number

DATA-TYP-FLOAT64 Converted to NCL floating point format. number.number

DATA-TYP-BOOL Converted to 0 or 1. 0 or 1

DATA-TYP-REAL-ENUM Not converted to external form unless
enumerated and in a shared token, in which case
it is converted to long with the high 16 bits as the
subsystem number.  (Removing the high 16 bits
converts to internal form.)

N.A.

DATA-TYP-BYTE-ENUM Not converted to external form unless
enumerated and in a shared token, in which case
it is converted to long with the high 16 bits as the
subsystem number.  (Removing the high 16 bits
converts to internal form.)

N.A.

DATA-TYP-STRUCT Fields are converted to external form using the
data types described in the preceding DATA-
TYP- rows.

N.A.

N.A. Not applicable

SPI to MDO Conversion Rule 2h.  If a message does not contain ZEMS-TKN-NODENAME,
NonStop NET/MASTER MS passes the value of ZEMS-TKN-SYSTEM to the Guardian
system procedure call GETSYSTEMNAME, which returns the system name associated
with the system number, and places the system name as the value of ZEMS-TKN-
NODENAME in the MDO variable.

For example, in a message that is read by the EMSREAD verb, if a system name of
\SYS1 is associated with a system number of 1, the MDO variable
&$EMS.SPI.TANDEM.ZEMS_TKN_NODENAME contains the value \SYS1.

SPI to MDO Conversion Rule 3

An EMS consumer distributor can create a token called ZEMS-TKN-PASSVAL.  This
occurs if the consumer distributor filter passes an event message with the PASS
statement.  If a message from an EMS consumer distributor contains ZEMS-TKN-
PASSVAL, NonStop NET/MASTER MS creates a second (repeated) occurrence of the
SPI element with the token and its value.

If the SPI element is repeated, you can use either of the following two variable names
to refer to the first occurrence of the SPI element in an MDO variable:

&$EMS.SPI
&$EMS.SPI{1}

You can use the following variable name to refer to ZEMS-TKN-PASSVAL in the
second occurrence of the SPI element in an MDO variable:

&$EMS.SPI{2}.TANDEM.ZEMS_TKN_PASSVAL

Detecting a Message With
SPI Tokens

After you read a message using the EMSREAD, INTREAD, LOGREAD, or MSGREAD
verb, you can easily detect whether it contains SPI tokens.  If a message does not
contain SPI tokens, the SPI element is not present in the message.  Conversely, if a
message does contain SPI tokens, the SPI element is present in the message.  The
following IF statement detects whether a message contains SPI tokens after it is read
by the EMSREAD verb:

IF &$EMS.SPI. \= "" THEN
   SAY SPI tokens are present in this message
ELSE
   SAY SPI tokens are not present in this message

Accessing SPI Tokens
From NCL

The following three NCL procedures show how you can easily access the data in the
SPI portion of an MDO variable.

The first NCL procedure is an EMSPROC procedure called ZEX1101N.  It is designed
to allow all messages that arrive at your NonStop NET/MASTER MS system to pass
through unchanged.

The second NCL is a MSGPROC procedure called ZEX1102N.  It is designed to trap
messages that contain SPI tokens and have TEST in the message text, and to analyze
and display their contents.

The third NCL procedure is an NCL procedure called ZEX1103N.  It uses the NCL
EMSSEND verb to send a message to your Event Management Service (EMS) collector.

To see how these NCL procedures work, follow these steps:

1. Type the first NCL procedure as shown, calling it ZEX1101N:

zex1101n: PROCEDURE
   /* EMSPROC procedure */
   SAY EMSPROC starting
   DO FOREVER
      EMSREAD
      EMSCONT
   END /*do*/
END zex1101n

2. Copy ZEX1101N to your customized procedure library.

3. Install ZEX1101N as your EMSPROC procedure using the following SYSPARMS
command from the OCS command input line:

SYSPARMS EMSPROC=ZEX1101N

Flushing your current EMSPROC procedure using the SYSPARMS
EMSPROC=FLUSH command achieves the same result, since this allows all EMS
messages to pass through.  You can use the SHOW SYSPARMS command or the
STATUS command to confirm that ZEX1101N is now your active EMSPROC
procedure.

4. Type the second NCL procedure as shown, calling it ZEX1102N:

zex1102n: PROCEDURE
   /* MSGPROC procedure to analyze and filter messages with */
   /* SPI tokens                                            */
   DO FOREVER
      /* Read message into a single variable */
      MSGREAD VARS=&message PARSE=NO
      /* See if message text contains TEST */
      &found = POS(TEST,&message)
      SELECT
         /* SPI tokens present and contains TEST */
         WHEN &$msg.spi \= "" AND &found > 0 THEN DO
            ASSIGN OPT=MDOELEMENTS VARS=&mdoelements*,
               FROM MDO=&$MSG.SPI.TANDEM
            SAY "Elements under &$MSG.SPI.TANDEM are"
            DO &i = 1 TO &SYS.VARCNT
               SAY &mdoelements&i is &$msg.spi.tandem.&mdoelements&i
               END
            MSGCONT COLOR=BLUE INTENS=LOW HLITE=BLINK
            END
         /* SPI tokens present but does not contain TEST */
         WHEN &$msg.spi \= "" AND &found <= 0 THEN
            MSGDEL
         OTHERWISE
            MSGCONT
         END
   END
END zex1102n

5. Install ZEX1102N as your active MSGPROC procedure using the following
command from the OCS command input line:

PROFILE MSGPROC=ZEX1102N

6. Type the third NCL procedure as shown, calling it ZEX1103N:

zex1103n: PROCEDURE
   /* Sends a message to the EMS collector */
   SAY "Sending a message to the EMS collector"
   ASSIGN MDO=&abc. MAP=$MSG
   &abc.text = TEST Hello world from &SYS.USER.ID
   &abc.msgattr.display.color = WHITE
   &abc.msgattr.display.hlite = HIGH
   &abc.spi.tandem.zspi_tkn_ssid = TANDEM.NNM.D30
   &abc.spi.tandem.zems_tkn_subject_mark = "TANDEM.ZEMS_TKN_TEXT{1}"
   &abc.spi.tandem.zems_tkn_text = TEST Hello world from &SYS.USER.ID
   &abc.spi.tandem.zems_tkn_eventnumber.# = 9999
   EMSSEND MDO=&abc.
END zex1103n

7. Execute the following commands to ensure that NonStop NET/MASTER MS can
receive EMS messages:

EMSDIST START
EMSCOLL START

The EMSDIST START command starts the EMS distributor interface.  The
EMSCOLL START command starts the EMS collector interface.  You can use the
STATUS command to confirm that the interfaces have started.

8. Execute the following commands to ensure that your OCS profile allows you to
receive EMS and unsolicited messages:

PROFILE EMS=YES UNSOL=YES MONMSG=YES

9. Execute ZEX1103N from the OCS command input line by using the following
command:

START ZEX1103N

10. Observe the results of installing ZEX1102N as your active MSGPROC procedure
and executing ZEX1103N, as shown in the following two screens:

Screen 1 of 2.

  (08:11) --------------------- OPERATOR CONTROL SERVICES ----------------------
   START ZEX1103N
   Sending a message to the EMS collector
   NNM1005 START ZEX1103N PROCESSING COMPLETE. NCLID 001072
   08:11:45 NNM9999 TEST Hello world from NMTJN
   Elements under &$MSG.SPI.TANDEM are
   ZSPI_TKN_HDRTYPE is 1
   ZSPI_TKN_CHECKSUM is 0
   ZSPI_TKN_LASTERR is
   ZSPI_TKN_LASTERRCODE is .. d
   ZSPI_TKN_MAX_FIELD_VERSION is 0
   ZSPI_TKN_SSID is TANDEM.NNM.D30
   ZSPI_TKN_USEDLEN is 127
   ZSPI_TKN_BUFLEN is 127
   ZEMS_TKN_CONSOLE_PRINT is 0
   ZEMS_TKN_CPU is 3
   ZEMS_TKN_CRTPID is $I003 3,130
   ZEMS_TKN_EMPHASIS is 0
   ZEMS_TKN_EVENTNUMBER is
   ZEMS_TKN_FORWARDED is 0
   ZEMS_TKN_GENTIME is 1992/06/14 22:11:44.602
   ZEMS_TKN_LOGTIME is 1992/06/14 22:11:44.613
 AUTO HOLD  ------------------ NonStop NET/MASTER D30 ---------------- --------
   M=>

Screen 2 of 2.

  (08:12) --------------------- OPERATOR CONTROL SERVICES ----------------------
   ZEMS_TKN_PIN is 130
   ZEMS_TKN_REDUNDANT is 0
   ZEMS_TKN_SUPPRESS_DISPLAY is 0
   ZEMS_TKN_SYSTEM is 250
   ZEMS_TKN_USERID is d
   ZEMS_TKN_NODENAME is \SYS1
   ZEMS_TKN_SUBJECT_MARK is TANDEM.ZEMS_TKN_TEXT{1}
   ZEMS_TKN_TEXT is TEST Hello world from NMTJN
   _____________________________________________________________________________
   ZSPI_TKN_MAX_FIELD_VERSION is 0
   ZSPI_TKN_SSID is TANDEM.NNM.D30
   ZSPI_TKN_USEDLEN is 127
   ZSPI_TKN_BUFLEN is 127
   ZEMS_TKN_CONSOLE_PRINT is 0
   ZEMS_TKN_CPU is 3
   ZEMS_TKN_CRTPID is $I003 3,130
   ZEMS_TKN_EMPHASIS is 0
   ZEMS_TKN_EVENTNUMBER is
   ZEMS_TKN_FORWARDED is 0
   ZEMS_TKN_GENTIME is 1992/06/14 22:11:44.602
   ZEMS_TKN_LOGTIME is 1992/06/14 22:11:44.613
 ---------- ------------------ NonStop NET/MASTER D30 ---------------- --------
   M=>

The EMSSEND verb is discussed in more detail in Section 17, “Developing System-
Level NCL Procedures.”

The Standard Map
$SEC

The standard map $SEC is used by the SECCALL verbs.  The SECCALL verbs—
SECCALL ADD, SECCALL CHANGE, and so on—manipulate user ID definition
records in the User ID Management Services (UMS) database.  The default map used
by these verbs to interpret the data in an MDO variable that contains a record being
written to or read from the User ID Management Services (UMS) database is $SEC.

Figure 11-3 shows the structure of the map $SEC.  It shows that data structured by this
map corresponds to the following MDO variable:

&stem-name.

stem-name.

specifies the name of the stem of the MDO variable that holds the record.

The name of a field in a record corresponds to MDO variables such as:

&stem-name.USERID
&stem-name.GROUP_USERID
&stem-name.NORM_UDEFN

If you are reading a record from the UMS database using SECCALL GET or SECCALL
QUERY, you do not have to use the ASSIGN verb to create the MDO variable that
holds the record.  Mapping Services automatically creates an MDO variable using the
stem name you specify after the MDO operand.  The following example reads a record
from the UMS database into the MDO variable &RECORD.:

SECCALL GET USERID=NNM001 MDO=&record.

If you are writing a record to the UMS database using SECCALL ADD, SECCALL
CHANGE, or SECCALL UPDATE, you can either use the ASSIGN verb to create the
MDO variable that holds the record before you store data in the MDO variable and
write it to the UMS database or you can use the MDO variable created by SECCALL
GET.  The following example creates an MDO variable, &ABC., using the ASSIGN
verb and writes it to the UMS database using the SECCALL ADD verb:

ASSIGN MDO=&abc. MAP=$SEC
…
/* store data in MDO variable */
…
SECCALL ADD USERID=NNM002 MDO=&abc.

The SECCALL verbs are discussed in more detail in Section 18, “Advanced NCL
Programming.”

Note Elements in an MDO variable that are not documented in Figure 11-3 are not currently supported.

Figure 11-3.  The Map $SEC (Page 1 of 4)

Figure 11-3.  The Map $SEC (Page 2 of 4)

Figure 11-3.  The Map $SEC (Page 3 of 4)

Figure 11-3.  The Map $SEC (Page 4 of 4)

User-Defined Maps NonStop NET/MASTER MS lets you create and use maps to access data structures
that you define.  These are called user-defined maps.  This subsection discusses the
following topics:

Why you would create a user-defined map

Creating a user-defined map.

Using a user-defined map.

Why You Would Create a
User-Defined Map

The standard maps supplied with NonStop NET/MASTER MS allow an NCL process
to work with data structures that are an integral part of NonStop NET/MASTER MS.
You do not need to define the data structures described by the $NCL, $MSG, $SEC, or
$SPI maps before an NCL verb can work with data described by one of these maps.

However, you may want an NCL process to work with a data structure that is not an
integral part of NonStop NET/MASTER MS.  That is, you may want an NCL process
to work with a data structure that you define.  The easiest way to do this is to create a
user-defined map, which defines the data structure.  Then you need to let the NCL
process know that the user-defined map is available for use.

NonStop NET/MASTER MS lets you create a user-defined map from a data structure
defined using the Tandem Data Definition Language (DDL).  DDL lets you create data
structures in a source schema file; you can use the DDL source schema file as the input
for creating a DDL data dictionary; you then use the DDL data dictionary as the input
for creating a user-defined map.

A DDL source schema file is typically used to create data structures for applications
external to NonStop NET/MASTER MS; these applications are typically written in
programming languages such as C, COBOL, FORTRAN, Pascal, TACL, and TAL.  The
ability to use the same DDL source schema file to create data structures for use by an
NCL process means that an NCL process and an external application can access the
same data.  For example, by using a user-defined map, an NCL process can:

Either read records from and write records to an Enscribe file that is created by an
external application, or write records to a new Enscribe file created by NonStop
NET/MASTER MS for later access by an external application.  This means that an
NCL process and an external application can share access to data in the same
Enscribe file.  For more information on accessing Enscribe files, see Section 12,
“Working With Files.”

Send messages to and receive messages from an external program, such as a
Pathway server class.  A user-defined map can be used to map the messages.  This
means that an NCL process can act as a PATHSEND requester to communicate
with an existing Pathway server class.  For more information on communicating
with Pathway server classes from an NCL process, see Section 15, “Working With
Pathway Server Classes.”

Creating a User-Defined
Map

To create a user-defined map, follow these steps:

1. Create a DDL source schema file using DDL statements.

2. Create a DDL data dictionary using the source schema file as input.

3. Create a map definition using the DDL data dictionary as input and add the map
definition to the NonStop NET/MASTER MS map file.

This subsection illustrates these steps by using a source schema file that defines data
structures for describing employee information.  The result of these steps is the
addition of a user-defined map, called EMPREC, to the map file.  This user-defined
map allows an NCL process to access substructures such as employee code, name,
address, birthday, sex, department, and pay period.

Note The following discussion does not attempt to illustrate the complexities of DDL.  It assumes that you know
how to use DDL and concentrates on the tasks you need to know to create a user-defined map.  For
detailed information on DDL, refer to the Data Definition Language (DDL) Reference Manual.

Step 1.  Creating a DDL Source Schema File

The first step in creating a user-defined map is creating a DDL source schema file.

The source schema file consists of DDL statements that define and describe DDL data
structures.  The source schema file is used as the input for creating a DDL data
dictionary, which is used as the input for creating a user-defined map.  The DDL
source schema file can contain both definitions (defined using the DDL DEFINITION
clause) and records (defined using the DDL RECORD clause); you can create maps
from either or both sources of input.

From a RECORD, you can create a map and the definition for creating a file, which
holds records with data structures that the map can interpret.  From a DEFINITION,
you can create a map only.  Therefore:

To read records from and write records to an Enscribe file that is created by an
external application, either a DEFINITION or RECORD is reasonable.

To write records to a new Enscribe file created by NonStop NET/MASTER MS for
later access by an external application, a RECORD is likely to be more applicable
input.

To send messages to and receive messages from an external program, such as a
Pathway server class, a DEFINITION is likely to be sufficient input.

The following example shows a source schema file, called $DATA2.UDDL.ZEX11051,
that defines data structures for describing employee information:

CONSTANT payperiod   VALUE IS 12.

DEF name.
    02 lastname      PIC X(12).
    02 firstname     PIC X(12).

    02 midinit       PIC X(1).
END

DEF address.
    02 street        PIC X(25).
    02 city          PIC X(14).
    02 state         PIC X(3).
    02 zip           PIC 9(5).
END

DEF date.
    02 day           PIC 9(2).
    02 month         PIC X(3).
    02 year          PIC 9(2).
END

DEF department.
    02 depnum.
       04 regnum     PIC 9.
       04 branchnum  PIC 9.
    02 deptype       TYPE ENUM.
       89 sales.
       89 marketing.
       89 admin.
       89 education.
       89 development.
END

DEF empdef.
    02 empcode       PIC 9(4).
    02 empname       TYPE name.
    02 empaddr       TYPE address.
    02 empbday       TYPE date.
    02 empsex        PIC X.
    02 empdep        TYPE department.
    02 emppay OCCURS payperiod TIMES
                     TYPE date.
END

RECORD emprec.
    FILE IS "$DATA2.UDDL.EMPFILE" KEY-SEQUENCED.
    DEF IS empdef.
    KEY IS emprec.empcode.
END

The source schema file has one constant (PAYPERIOD) with a value of 12.  This is used
in the definition of the EMPPAY field to determine the number of occurrences of the
field.

The schema defines five definitions (NAME, ADDRESS, DATE, DEPARTMENT, and
EMPDEF).  The NAME, ADDRESS, and DATE groups are defined with fields at the
same level.  The DEPARTMENT group is defined with one group (DEPNUM) and one

field (DEPTYPE), each of which is at the same level.  The DEPNUM group has two
fields (REGNUM and BRANCHNUM).  The DEPTYPE field defines an enumerated
data type.

The EMPDEF group is defined using four previously defined data structures (NAME,
ADDRESS, DATE, and DEPARTMENT).  DATE is used twice:  first, for EMPBDAY;
second, for EMPPAY.  The value of PAYPERIOD determines the number of
occurrences (12) of the EMPPAY field.

The schema defines one record (EMPREC).  This defines the data structure of records
in a key-sequenced file ($DATA2.UDDL.EMPFILE), which is the database file that
holds employee records.  The data structure is defined using the EMPDEF group.

Step 2.  Creating a DDL Data Dictionary

The second step in creating a user-defined map is creating a DDL data dictionary from
the DDL source schema file.  The DDL data dictionary consists of 14 prenamed and
predefined files.  The DDL data dictionary is used as the input for creating a
user-defined map.

For example, the following command runs the DDL compiler to create two types of
output:  a DDL data dictionary and output that FUP can use to create an Enscribe file:

DDL /IN ZEX11051/ DICT, FUP FUPSRC

The following listing shows partial results of the command, which is executed from
the $DATA2.UDDL subvolume:

            …
    0       ?DICT, FUP FUPSRC
            Dictionary created on subvol $DATA2.UDDL.
             *** WARNING *** This dictionary is NOT audited
            Dictionary opened on subvol $DATA2.UDDL for update access.
            Output source for FUP is opened on $DATA2.UDDL.FUPSRC
            …
   45       RECORD emprec.
   46           FILE IS "$DATA2.UDDL.EMPFILE" KEY-SEQUENCED.
   47           DEF IS empdef.
   48           KEY IS emprec.empcode.
   49       END
            Record EMPREC size is 172 bytes.
            Record EMPREC added to dictionary.
            FUP output produced for EMPREC.
            …

Of most significance in this partial listing, because it is going to be used as the input to
create the EMPREC user-defined map later in this section, is that the record EMPREC
is added to the DDL data dictionary.

The listing also shows that FUP output is produced.  FUP can use this output to create
a key-sequenced Enscribe file called $DATA2.UDDL.EMPFILE to contain employee
information that is mapped by the EMPREC map.  Using this file, an NCL process and
an external application can easily share access to the same employee data.  The
following listing shows the FUP output:

< SCHEMA PRODUCED DATE - TIME :10/20/92 13:58:23
< SECTION EMPREC
< Record EMPREC created on 10/20/92 at 13:58
RESET
   SET TYPE K
   SET KEYLEN 4
   SET REC 172
   SET BLOCK 4096
   SET IBLOCK 4096
   SET EXT( 4, 32 )
   SET MAXEXTENTS 100
CREATE $DATA2.UDDL.EMPFILE

The following FUP command uses the FUP output to create
$DATA2.UDDL.EMPFILE:

FUP /IN FUPSRC/

Step 3.  Creating and Adding a Map Definition

The third step in creating a user-defined map is creating a map definition using the
data structures defined in the DDL data dictionary, and adding the map definition to
the map file.  The MDMAINT utility program, which is supplied with NonStop
NET/MASTER MS, is used to perform these system management tasks.  For further
information on the MDMAINT utility program, refer to the NonStop NET/MASTER MS
System Management Guide.

Assume NonStop NET/MASTER MS was installed with a process character of Z on
$DATA2.ZNNM on your Tandem system.  The following TACL RUN command
invokes the MDMAINT program to create a map definition for the record EMPREC
and add it to the map file:

RUN $DATA2.ZNNM.MDMAINT /PRI 140, NAME $ZMDM/
    $DATA2.ZNNMDATA.MAPFILE
    ADD $DATA2.UDDL EMPREC R

After the user-defined map is added to the map file, an NCL process can use the map
to access data.  For example, after the EMPREC map definition is added to the
NonStop NET/MASTER MS data dictionary, an NCL process can use the map to
access employee information.

You can determine whether a map is available for use by using the SHOW MAPS
command.  The following screen shows that, in addition to the standard maps, the
EMPREC map is also available for use:

  (15:20) --------------------- OPERATOR CONTROL SERVICES ----------------------
   SHOW MAPS
   NNM0256 MAP-NAME     MAP-KEY   USE-CNT  STATUS
   NNM0257 $MAP         00000030        0  NAMEONLY
   NNM0257 $MSG         00000002        1  LOADED
   NNM0257 $NCL         00000001        0  NAMEONLY
   NNM0257 $SEC         00000010        1  LOADED
   NNM0257 $SPI         00000077        1  LOADED
   NNM0257 EMPREC       80000000        0  NAMEONLY
   NNM0999 *END* 6 MAPS DISPLAYED
   _____________________________________________________________________________

 ---------- ------------------ NonStop NET/MASTER D30 ---------------- --------
   M=>

Note The display does not show the user-defined map until it is loaded into memory.  If NonStop NET/MASTER
MS is not running and its map file is closed, this occurs automatically when NonStop NET/MASTER MS is
started.  If NonStop NET/MASTER MS is running and its map file is open, this occurs when an NCL
process refers to the map or you execute the SYSPARMS MAPLOAD command.

Before you attempt to use the user-defined map, it is useful to confirm that the data
structures defined using DDL correspond to the data structures you expect in the
user-defined map.  The following NCL procedure lets you display the groups and
fields in a user-defined map.

zex1104n: procedure
   /* Displays a user-defined map */
   ASSIGN MDO=&x. MAP=&1
   CALL show_map( 0, "" ) SHARE &x.
   show_map: PROCEDURE
      &nspaces = &1
      &piece = &2
      &spaces = COPIES( " ", &nspaces )
      IF &piece \== "" THEN
         &piecedot = &piece"."
      ASSIGN OPT=MAPFIELDS ARGS FROM MDO=&x.&piece
      DO &i = 1 TO &SYS.VARCNT
         &y = &piecedot || &&i
         ASSIGN OPT=MAPFIELDS VARS=&var FROM MDO=&x.&y
         IF &SYS.VARCNT = 0 THEN &type = Field
            ELSE &type = Group
         SAY &spaces || &type &&i
         CALL show_map( &nspaces + 2, &y ) SHARE &x.
     END /*do*/
   end show_map
end zex1104n

The name of the map is entered as the first parameter.

The following screen shows the results of executing the procedure to display the
groups and fields in the EMPREC map (EMPREC is entered as the first parameter):

  (15:25) --------------------- OPERATOR CONTROL SERVICES ----------------------
   START ZEX1104N EMPREC
   Group EMPADDR
     Field CITY
     Field STATE
     Field STREET
     Field ZIP
   Group EMPBDAY
     Field DAY
     Field MONTH
     Field YEAR
   Field EMPCODE
   Group EMPDEP
     Group DEPNUM
       Field BRANCHNUM
       Field REGNUM
     Field DEPTYPE
     Field DEPTYPE#
   Group EMPNAME
     Field FIRSTNAME
     Field LASTNAME
     Field MIDINIT
 MSG QUEUED ------------------ NonStop NET/MASTER D30 ---------------- --------
   M=>

Note A user-defined map does not have elements, only groups and fields.

Using a User-Defined Map After you have created a user-defined map, an NCL process can use the map.  An NCL
process uses a user-defined map as it uses a standard map.  That is, it uses the names
of substructures within the map to locate data at a certain position in an MDO
variable.  Section 10, “Working with Mapping Services,” explains how to use MDO
variables.

The following NCL procedure uses the EMPREC map to write a record to and then
read the record from $DATA2.UDDL.EMPFILE.  The NCL procedure assumes that
you have created the EMPREC map and the $DATA2.UDDL.EMPFILE, as described
earlier in this section.  (See Section 12, “Working With Files,” for more information on
the FILE verbs, which are used in this NCL procedure.)

To see how this NCL procedure works, follow these steps:

1. Type the NCL procedure as shown, calling it ZEX1105N:

zex1105n: PROCEDURE
   /* Opens a mapped key-sequenced file (which is   */
   /* mapped using a user-defined map), adds a      */
   /* record, gets the record, displays the record, */
   /* then closes the file.                         */
   ON ERROR FLUSH
   &usermap = EMPREC
   &id      = EMPFILE
   CALL open_file SHARE &usermap,&id
   IF &SYS.FILE.RC = 8 THEN
         CALL add_record SHARE &usermap
      ELSE DO
         SAY "Flushing after OPEN_FILE"
         FLUSH
         END /*do*/
   IF &SYS.FILE.RC = 0 THEN
         CALL get_record SHARE &usermap,&getmdo.
      ELSE DO
         SAY "Flushing after ADD_RECORD"
         FLUSH
         END /*do*/
   IF &SYS.FILE.RC = 0 THEN
         CALL display_record( 0, "" ) SHARE &usermap,&getmdo.
      ELSE DO
         SAY "Flushing after GET_RECORD"
         FLUSH
         END /*do*/
   CALL close_file SHARE &id
   EXIT
   /* ====================  OPEN FILE  ==================== */
   open_file: PROCEDURE SHARE &usermap,&id
      /* This procedure assumes that the file exists and    */
      /* has been opened for access by NonStop NET/MASTER.  */
      /* Therefore, this procedure only opens the file for  */
      /* access by NCL.  (You can use FUP to create a key-  */

      /* sequenced file.  The UDBCTL OPEN command opens a   */
      /* file for access by NonStop NET/MASTER.)            */
      /* It is recommended that KEYEXTR is equal to NO for  */
      /* files mapped with a user-defined map.              */
      FILE OPEN FORMAT=MAPPED MAP=&usermap ID=&id KEYEXTR=NO
      SELECT &SYS.FILE.RC
         WHEN 0  THEN SAY "0 Read-only access"
         WHEN 4  THEN SAY "4 Read and write access"
         WHEN 8  THEN SAY "8 Read, write, and delete access"
         WHEN 12 THEN SAY "12 No access"
         WHEN 16 THEN SAY "16 "&SYSMSG
         OTHERWISE
            SAY Unexpected error
      END /*select*/
      EXIT &SYS.FILE.RC
   END open_file
   /* ==================== ADD RECORD ==================== */
   add_record: PROCEDURE SHARE &usermap
      /* Adds one record, mapped by a user-defined map. */
      ASSIGN MDO=&addmdo. MAP=&usermap
      &addmdo.empcode                  = 77
      &addmdo.empname.lastname         = Smith
      &addmdo.empname.firstname        = John
      &addmdo.empname.midinit          = ABC
      &addmdo.empaddr.street           = 99 Spring St
      &addmdo.empaddr.city             = Mycity
      &addmdo.empaddr.state            = Mystate
      &addmdo.empaddr.zip              = 12345
      &addmdo.empbday                  = 15May59
      &addmdo.empsex                   = m
      &addmdo.empdep.depnum.regnum     = 9
      &addmdo.empdep.depnum.branchnum  = 5
      &addmdo.empdep.deptype           = education
      &addmdo.emppay{1}.day            = 31
      &addmdo.emppay{1}.month          = Jan
      &addmdo.emppay{1}.year           = 99
      &addmdo.emppay{2}                =
      &addmdo.emppay{3}                = 31Mar99
      &addmdo.emppay{4}                =
      &addmdo.emppay{5}                = 31May99
      &addmdo.emppay{6}                =
      &addmdo.emppay{7}                = 31Jul99
      &addmdo.emppay{8}                =
      &addmdo.emppay{9}                = 30Sep99
      &addmdo.emppay{10}               =
      &addmdo.emppay{11}               = 30Nov99
      &addmdo.emppay{12}               =
      FILE PUT MDO=&addmdo. KEY=&addmdo.empcode
      SELECT &SYS.FILE.RC
         WHEN 0  THEN DO
            SAY "0 Record added"
            SAY "PUT KEY operand value is |"&addmdo.empcode"|"
            SAY "PUT &SYS.FILE.KEY value is |"&SYS.FILE.KEY"|"
            END /*do*/
         WHEN 4  THEN SAY "4 Record exists, not added"
         WHEN 8  THEN SAY "8 Error "&SYS.FILE.ERROR

         WHEN 16 THEN SAY "16 "&SYSMSG
         OTHERWISE
            SAY Unexpected error
      END /*select*/
      EXIT &SYS.FILE.RC
   END add_record
   /* ==================== GET RECORD ==================== */
   get_record: PROCEDURE SHARE &usermap,&getmdo.
      /* Gets one record, mapped by a user-defined map. */
      &key = "77  "
      FILE GET KEY=&key MDO=&getmdo.
      SELECT &SYS.FILE.RC
         WHEN 0  THEN DO
            SAY "0 Record retrieved"
            SAY "GET KEY operand value is |"&key"|"
            SAY "GET &SYS.FILE.KEY value is |"&SYS.FILE.KEY"|"
            END /*do*/
         WHEN 4  THEN SAY "4 Record not found or EOF"
         WHEN 8  THEN SAY "8 Error "&SYS.FILE.ERROR
         WHEN 16 THEN SAY "16 "&SYSMSG
         OTHERWISE
            SAY Unexpected error
      END /*select*/
      EXIT &SYS.FILE.RC
   END get_record
   /* ================== DISPLAY RECORD ================== */
   display_record: PROCEDURE SHARE &usermap,&getmdo.
      /* Displays one record, mapped by a user-defined map. */
      /* Distinguishes between DDL groups and fields.       */
      &nspaces = &1
      &piece = &2
      &spaces = COPIES( " ", &nspaces )
      IF &piece \== "" THEN
         &piecedot = &piece"."
      ASSIGN OPT=MDOFIELDS ARGS FROM MDO=&getmdo.&piece
      DO &i = 1 TO &SYS.VARCNT
         &y = &piecedot || &&i
         ASSIGN OPT=MDOFIELDS VARS=&var FROM MDO=&getmdo.&y
         IF &SYS.VARCNT = 0 THEN &type = Field
            ELSE &type = Group
         SAY &spaces || &type &&i "is |"&getmdo.&y"|"
         CALL display_record( &nspaces + 2, &y ) SHARE &getmdo.
         END /*do*/
   END display_record
   /* ==================== CLOSE FILE ==================== */
   close_file: PROCEDURE SHARE &id
      /* Close file for access by NCL, but not for access */
      /* by NonStop NET/MASTER.                           */
      FILE CLOSE ID=&id
   END close_file
END zex1105n

Note You should use the SHOW UDB command to ensure that EMPFILE is not being used as a UDB identifier
at your installation.  If it is, edit the NCL procedure so that it uses a different UDB identifier.



User-Defined Maps

Standard and User-Defined Maps

11–38 106160 Tandem Computers Incorporated

This NCL procedure shows how to refer to data structures in the MDO variable
mapped by EMPREC and how to assign data to the groups and fields in the MDO
variable.

A full name hierarchy is required to refer to a field.  Assigning data to a field
explicitly assigns data only to that field.  The following example assigns the value
Smith to the field &ADDMDO.EMPNAME.LASTNAME:

&ADDMDO.EMPNAME.LASTNAME = Smith

A full name hierarchy is required to refer to a group.  Assigning data to a group
implicitly assigns data to the fields that are subordinate to the group.  The
following example assigns the value 15May59 to the group
&ADDMDO.EMPBDAY:

&ADDMDO.EMPBDAY = 15May59

This assignment statement has the same result as the following assignment
statements:

&ADDMDO.EMPBDAY.DAY = 15
&ADDMDO.EMPBDAY.MONTH = May
&ADDMDO.EMPBDAY.YEAR = 59

An integer index is required to refer to a specific repeated group or field.  The
following example assigns values to &ADDMDO.EMPPAY, which is a repeated
group:

&ADDMDO.EMPPAY{1}.DAY = 31
&ADDMDO.EMPPAY{1}.MONTH = Jan
&ADDMDO.EMPPAY{1}.YEAR = 99
&ADDMDO.EMPPAY{3} = 31Mar99

The number sign is omitted to refer to the unenumerated form of an enumerated
field.  The following example assigns the value education to
&ADDMDO.EMPDEP.DEPTYPE:

&ADDMDO.EMPDEP.DEPTYPE = education

2. Execute the following command from the OCS command input line:

UDBCTL OPEN=$DATA2.UDDL.EMPFILE ID=EMPFILE

This command opens $DATA2.UDDL.EMPFILE for use by NonStop
NET/MASTER MS.

Note You should use the SHOW UDB command to ensure that EMPFILE is not being used as a UDB identifier
at your installation.  If it is, use a different UDB identifier in the UDBCTL OPEN command.

3. Execute the following command from the OCS command input line:

START ZEX1105N

4. Observe the results, as shown in the following two screens:

Screen 1 of 2

  (11:58) --------------------- OPERATOR CONTROL SERVICES ----------------------
   UDBCTL OPEN=$DATA2.UDDL.EMPFILE ID=EMPFILE
   NNM0300 OPEN REQUEST COMPLETE
   START ZEX1105N
   8 Read, write, and delete access
   0 Record added
   PUT KEY operand value is |77  |
   PUT &SYS.FILE.KEY value is |77  |
   0 Record retrieved
   GET KEY operand value is |77  |
   GET &SYS.FILE.KEY value is |77  |
   Group EMPADDR is |99 Spring St             Mycity        Mys12345|
     Field CITY is |Mycity        |
     Field STATE is |Mys|
     Field STREET is |99 Spring St             |
     Field ZIP is |12345|
   Group EMPBDAY is |15May59|
     Field DAY is |15|
     Field MONTH is |May|
     Field YEAR is |59|
   Field EMPCODE is |77  |
   Group EMPDEP is |95 .|
 MSG QUEUED ------------------ NonStop NET/MASTER D30 ---------------- --------
   M=>

Screen 1 shows that the UDBCTL OPEN command successfully opens
$DATA2.UDDL.EMPFILE for access by NonStop NET/MASTER MS.  Return
codes of 8, 0, and 0 indicate that the FILE OPEN, FILE PUT, and FILE GET verbs
successfully perform their respective operations.  The record key for the single
record that is added (FILE PUT) and retrieved (FILE GET) from the file is “77  “;
that is, 77 followed by two spaces (not nulls).

Screen 1 displays the value and length (using vertical bars (|)) of the data in the
groups and the fields in the record.  It shows that data that exceeds the length
defined in the DDL source schema file is truncated to the correct length.  For
example, Mystate (seven bytes) is truncated to Mys (three bytes).  It shows that
data that is less than the length defined in the DDL source schema file is padded
with spaces to the correct length.  For example, Mycity (six bytes) is padded to
“Mycity        “ (14 bytes).

Screen 1 shows the difference between groups and fields.  A group can contain
other groups or fields, but a field is the most fundamental unit in a mapped data
structure.  It demonstrates two ways to access and display data in groups and
fields.  Referring to a group name refers to the data in all the fields that belong to
the group:  for example, referring to EMPBDAY obtains the value 15May59 (the
fields are DAY, MONTH, and YEAR).  Referring to a field name refers only to the
data in the specific field.  For example, referring to DAY obtains the value 15.

Screen 2 of 2

  (11:58) --------------------- OPERATOR CONTROL SERVICES ----------------------
     Group DEPNUM is |95|
       Field BRANCHNUM is |5|
       Field REGNUM is |9|
     Field DEPTYPE is |EDUCATION|
     Field DEPTYPE# is |3|
   Group EMPNAME is |Smith       John        A|
     Field FIRSTNAME is |John        |
     Field LASTNAME is |Smith       |
     Field MIDINIT is |A|
   Group EMPPAY{1} is |31Jan99|
     Field DAY is |31|
     Field MONTH is |Jan|
     Field YEAR is |99|
   Group EMPPAY{2} is |       |
     Field DAY is |  |
     Field MONTH is |   |
     Field YEAR is |  |
   Group EMPPAY{3} is |31Mar99|
     Field DAY is |31|
     Field MONTH is |Mar|
     Field YEAR is |99|
 AUTO HOLD  ------------------ NonStop NET/MASTER D30 ---------------- --------
   M=>

Screen 2 has features in addition to those described for Screen 1.  Screen 2 shows
how enumerated and unenumerated values are displayed.  For example,
DEPTYPE refers to an unenumerated value (EDUCATION) and DEPTYPE# refers
to an enumerated equivalent (3).  Other enumerated values are sales (0), marketing
(1), admin (2), and development (4).

Screen 2 shows how repeated groups are displayed.  For example, EMPPAY is
repeated 12 times.  The first, third, fifth, seventh, ninth, and eleventh occurrences
have pay information.  The second, fourth, sixth, eighth, tenth, and twelfth fields
have only blanks.

5. Execute the following command from the OCS command input line:

UDBCTL CLOSE=$DATA2.UDDL.EMPFILE

This command closes $DATA2.UDDL.EMPFILE for use by NonStop
NET/MASTER MS.

Considerations for User-
Defined Maps

The following considerations apply when creating a DDL source schema file to
subsequently create a user-defined map:

A user-defined map name must be from 1 through 32 characters long.  You can use
the following characters in a user-defined map name, as long as the first character
is not a digit:

Alphabetic characters:  A-Z and a-z
Numeric characters:  0 through 9
Other characters:  # $ @

The following DEFINITION and RECORD clauses are supported by NonStop
NET/MASTER MS:

OCCURS
PIC[TURE]
TACL
TYPE data-type
TYPE def-name
89 enumeration

The following DEFINITION and RECORD clauses are not supported by NonStop
NET/MASTER MS:

AS
CONSTANT
DISPLAY
EDIT-PIC
FILLER
HEADING
HELP
KEYTAG
MUST BE
NULL
OCCURS … DEPENDING ON …
REDEFINES
SPI-NULL
UPSHIFT
USAGE
[NO]VALUE
66 RENAMES
88 condition-name

You can create an Enscribe entry-sequenced, key-sequenced, relative, or
unstructured file using DDL, but NCL does not support access to relative or
unstructured files (except edit files).  Section 12, “Working With Files,” discusses
Enscribe files in more detail.

A key-sequenced file created using DDL has a key.  When you open the file by
using the FILE OPEN verb or set the current characteristics of the file using the
FILE SET verb, you must specify KEYEXTR=NO.  Otherwise, when you are
adding a record to the file using the FILE ADD or FILE PUT verb, the current
value of the &SYS.FILE.KEY system variable is inserted before the key you have
defined using DDL (to which you have assigned a value in the NCL process).  See
Section 12, “Working With Files,” for more information on the affect of the
KEYEXTR operand.



User-Defined Maps

Standard and User-Defined Maps

11–42 106160 Tandem Computers Incorporated

The following steps show what you should do to work with a file that you create
using DDL using its corresponding map:

1. Assign a key for the key-sequenced file as part of the RECORD clause, for
example:

DEF empdef
    02 empcode  PIC 9(4).
    …
END

RECORD emprec.
    FILE IS "$DATA2.UDDL.EMPFILE" KEY-SEQUENCED.
    DEF IS empdef.
    KEY IS emprec.empcode.
END

2. Create the map, add to the map file, create the key-sequenced file, and open
the key-sequenced file for access by NonStop NET/MASTER MS, for example:

UDBCTL OPEN=$DATA2.U DDL.EMPFILE ID=EMPFILE

3. Open the file from an NCL process, specifying KEYEXTR=NO, for example:

FILE OPEN FORMAT=MAPPED MAP=EMPREC ID=EMPFILE KEYEXTR=NO

When you are creating a record to write to the file, you should follow these steps:

1. Create the MDO variable that is to contain the record by using the ASSIGN
verb with the specified map, for example:

ASSIGN MDO=&addmdo. MAP=EMPREC

2.  Assign data to the fields in the MDO variable, for example:

&addmdo.empname.lastname = Smith
&addmdo.empname.firstname = John
&addmdo.empname.midinit = ABC
&addmdo.empaddr.street = 99 Spring St
&addmdo.empaddr.city = Mycity
&addmdo.empaddr.state = Mystate
&addmdo.empaddr.zip = 12345

3. Assign the value of the record key to the MDO variable that corresponds to
the key you defined using DDL, for example:

&addmdo.empcode = 7777

4. Specify this MDO variable as the value of the key on the FILE ADD or FILE
PUT verb, for example:

FILE PUT MDO=&addmdo. KEY=&addmdo.empcode

The value of &SYS.FILE.KEY also has the value specified by this variable.  In this
example, the value of &SYS.FILE.KEY is 7777.

If you do not use the KEY operand in the FILE ADD or FILE PUT verb, the value
of the key on the file is the current value of &SYS.FILE.KEY rather than the value
in the MDO variable that corresponds to the key you defined using DDL.  If the
value of &SYS.FILE.KEY is null, a run-time error is raised.

When you are reading from an existing file, the value of &SYS.FILE.KEY and the
value of the MDO variable that corresponds to the key you defined using DDL are
the same.

NonStop NET/MASTER MS converts some TACL data types from internal form
to external form before being stored in an MDO variable.  Table 11-5 lists the data
types that are converted to external form.

Table 11-5.  Considerations for TACL Data Types

TACL Data Type Conversion Method Used Result of Conversion to External Form

CRTPID Converted to external form using the Guardian
FNAMECOLLAPSE procedure call.  (FNAMEEXPAND
converts to internal form.)

\sys.$process-name cpu,pin or
\sys.cpu,pin cpu,pin

DEVICE Converted to external form using the Guardian
FNAMECOLLAPSE procedure call.  (FNAMEEXPAND
converts to internal form.)

\sys.$device

FNAME Converted to external form using the Guardian
FNAMECOLLAPSE procedure call.  (FNAMEEXPAND
converts to internal form.)

$volume.subvol.filename (local filename) or
\sys.$volume.subvol.filename (network filename)

FNAME 32 Converted to external form using the Guardian
FNAME32COLLAPSE procedure call.
(FNAME32EXPAND converts to internal form.)

\sys.$volume.subvol.filename

SSID Converted to external form using the Guardian
SSIDTOTEXT procedure call.  (TEXTTOSSID converts to
internal form.)

owner.ss.version or
0.0.0

SUBVOL Converted to external form using the Guardian
FNAMECOLLAPSE procedure call.  (FNAMEEXPAND
converts to internal form.)

\sys.$volume.subvol

TSTAMP Converted to the following time format:  yyyy/mm/dd
hh:mm:ss.ddd (in text).

yyyy/mm/dd hh:mm:ss.ddd

TRANSID Converted to external form using the Guardian
TRANSIDTOTEXT procedure call.  (TEXTTOTRANSID
converts to internal form.)

\sys-name(crash-count).cpu.sequence,
\sys-number(crash-count).cpu.sequence,
\sys-name.cpu.sequence,
\sys-number.cpu.sequence or
cpu.sequence

USERNAME N.A.  Treated as a string. N.A.

N.A. Not applicable

NonStop NET/MASTER MS stores some TYPE clause data types in internal form
different from how they are stored in external form in an MDO variable.
Table 11-6 summarizes how TYPE clause data types are treated.

Table 11-6.  Considerations for TYPE Clause Data Types

TYPE Clause Data Type Internal Form External Form

BINARY 8 N.A. N.A.

BINARY 8 UNSIGNED N.A. N.A.

BINARY 16 N.A. N.A.

BINARY 16 UNSIGNED N.A. N.A.

BINARY 16, n Stored as an integer without the decimal point.
The value of the integer must be from -32768
through 32767.  Example:  1532.

Stored as a sequence of digits with a decimal
point.  If negative, the digits are preceded by the
negative unary operator (-).  Example:  15.32
(assuming n=2).

BINARY 16, n UNSIGNED Stored as an unsigned integer without the
decimal point.  The value of the integer must be
from 0 through 65535.  Example:  1532.

Stored as a sequence of digits with a decimal
point.  Example:  15.32 (assuming n=2).

BINARY 64 N.A. N.A.

BINARY 64, n Stored as an integer without the decimal point.
The value of the integer must be from -32768
through 32767.    Example:  1532.

Stored as a sequence of digits with a decimal
point.    Example:  15.32 (assuming n=2).

COMPLEX The two numbers are stored in eight bytes.  Each
number is a 4-byte FLOAT 32 value.

Each number is stored as a sequence of digits.
The numbers are surrounded by parentheses
(( )) and separated by a comma (,).  Example:
(10,15).

ENUM Stored as an integer.  Example:  0. The unenumerated value is stored as a
sequence of digits.  Example:  0.  The
enumerated value is stored as a sequence of
characters.  Example:  RED.  (You refer to the
enumerated value by appending a number sign
(#) to the variable name.)

FLOAT N.A. N.A.

FLOAT 32 N.A. N.A.

FLOAT 64 N.A. N.A.

LOGICAL Stored as an integer.  Example:  5. If the value in internal form is 0, then the value in
external form is 0; otherwise, 1.  Example:  1.

LOGICAL 2 Stored as an integer.  Example:  0. If the value in internal form is 0, then the value in
external form is 0; otherwise, 1.  Example:  0.

LOGICAL 4 Stored as a long.  Example:  1. If the value in internal form is 0, then the value in
external form is 0; otherwise, 1.  Example:  1.

N.A. Not applicable

The following table shows some examples of values that are valid for the specified
TYPE clause:

TYPE Clause Original Value Reason

BINARY 16, 2 123.45 The value of 12345 is greater than -32768
and less than 32767.

BINARY 16, 2 UNSIGNED 11.11 The value of 1111 is greater than 0 and less
than 65535.

BINARY 64, 1 756.32 After 756.32 is rounded down to 756.30, the
value of 7563 is greater than -32768 and
less than 32767.

The following table shows some examples of values that are invalid for the
specified TYPE clause:

TYPE Clause Original Value Reason

BINARY 16, 2 756.32 The value of 75632 is greater than 32767.

BINARY 16, 2 7563.26 After 7563.26 is rounded up to 7563.30, the
value of 756330 is greater than 32767.

BINARY 16, 3 UNSIGNED 756.32 The value of 756320 is greater than 65535.

BINARY 64, 4 756.32 The value of 7563200 is greater than 32767.

Table 11-7 summarizes how NonStop NET/MASTER MS treats PICTURE clause
data types.

Table 11-7.  Considerations for PICTURE Clause Data Types

PICTURE Clause Data Type Internal Form External Form

A Stored as a sequence of characters.  Example:
ABC.

Stored as a sequence of characters.  Example:
ABC.

X Stored as a sequence of characters.  Example:
ABC.

Stored as a sequence of characters.  Example:
ABC.

9 Stored as a sequence of digits.  Example:  123. Stored as a sequence of digits.  Example:  123.

S9[…] Stored as a sequence of digits.  Example:  123. Stored as a sequence of digits.  Example:  123.

S9[…]V9(…) Stored as a sequence of digits.  Example:  -234. Stored as a sequence of digits with a decimal
point.  Example:  -2.34.

T9[…] N.A. Stored as a sequence of characters.

9[…]S N.A. Stored as a sequence of characters.

9[…]T N.A. Stored as a sequence of characters.

N.A. Not applicable

